Solid Waste Disposal

Two cities use incinerators to reduce waste before placing it in landfills.

© Dennis Bricker
Dept of Mechanical & Industrial Engineering
The University of Iowa

City 1 produces **500** tons of waste per day, and
City 2 produces **400** tons of waste per day.



Waste must be incinerated at **incinerator 1** or **incinerator 2.**

Each incinerator can process up to **500** tons of waste per day.

The cost to incinerate waste is $**40**/ton at incinerator 1 and $**30**/ton at incinerator 2.

Incineration reduces each ton of waste to **0.2** tons of debris, which must be dumped at one of **two landfills**.

Each landfill can receive at most **200** tons of debris per day.

It costs $3 per mile to transport a ton of material (either debris or waste).

Distances (in miles) between locations are:

	Incinerator #1	Incinerator #2
City# 1	**30**	5
City #2	36	42

	Landfill #1	Landfill #2
Incin #1	**5**	**8**
Incin #2	**9**	**6**

Formulate an LP that can be used to minimize the total cost of disposing of the waste of both cities.

Definition of variables

Xij = tons/day of City #i waste that is sent to Incinerator #j
i=1,2; j=1,2

Yjk = tons/day of debris sent from Incinerator #j to Landfill #k
j=1,2; k=1,2

(There are a total of 8 variables.)





LP Model formulation

$$\text{MIN } 130\,X_{11} + 45\,X_{12} + 148\,X_{21} + 156\,X_{22} + 15\,Y_{11} + 24\,Y_{12} + 27\,Y_{21} + 18\,Y_{22}$$

SUBJECT TO

$$X_{11} + X_{12} = 500$$
$$X_{21} + X_{22} = 400$$
$$X_{11} + X_{21} <= 500$$
$$X_{12} + X_{22} <= 500$$
$$0.2(X_{11} + X_{21}) = Y_{11} + Y_{12}$$
$$0.2(X_{12} + X_{22}) = Y_{21} + Y_{22}$$
$$Y_{11} + Y_{21} <= 200$$
$$Y_{12} + Y_{22} <= 200$$
$$X_{ij} >= 0 \;,\; Y_{ij} >= 0 \quad \text{For all } i=1,2 \;\&\; j=1,2$$

LINDO output

```
        OBJECTIVE FUNCTION VALUE
   1)   84700.0000

VARIABLE        VALUE          REDUCED COST
     X12      500.000000           .000000
     X21      400.000000           .000000
     Y11       80.000000           .000000
     Y22      100.000000           .000000

     ROW   SLACK OR SURPLUS     DUAL PRICES
      2)         .000000        -48.600000
      3)         .000000       -151.000000
      4)         .000000        -15.000000
      5)         .000000        -18.000000
      6)      120.000000           .000000
      7)      100.000000           .000000
      8)      100.000000           .000000
      9)         .000000           .000000
```


Thus, City #1 sends its waste to Incinerator #2,

& City #2 sends its waste to Incinerator #1.

Incinerator #1 sends all its debris to Landfill #1

& Incinerator #2 sends all its debris to Landfill #2.



Use a modeling language (e.g. LINGO or MPL) to formulate this model!



```
MODEL: ! WASTE INCINERATION & DISPOSAL IN LANDFILLS;
SETS:  ! DECISION VARIABLES ARE X, Y, & Z
         X = TRANSPORT FROM CITIES TO INCINERATORS
         Z = WASTE BURNED IN INCINERATOR
         Y = TRANSPORT FROM INCINERATORS TO LANDFILLS;
   CITIES/1..2/: OUTPUT;
   INCINERATORS/1..2/: ICOST, ICAP, Z;
   LANDFILLS/1..2/: LCAP;
   PICKUP(CITIES,INCINERATORS): DX, X;
   DISPOSE(INCINERATORS,LANDFILLS): DY, Y;
ENDSETS
```

Notes: be sure to end comments with a semicolon!

Variable Z is defined for convenience only


```
DATA:
   OUTPUT= 500 400;
   ICOST =   40 30;
   ICAP  = 500 500;
   LCAP  = 200 200;
   DX    =  30   5   !  DISTANCES BETWEEN CITIES ;
            36  42;  !     & INCINERATORS;
   DY    =   5   8   !  DISTANCES BETWEEN INCINERATORS;
              9   6;  !     & LANDFILLS;
   COSTPERMI = 3.0 ; ! TRANSPORTATION COST PER MILE;
   REDUCTION = 0.2 ; ! REDUCTION FACTOR:WASTE TO DEBRIS;
ENDDATA
```


The mathematical statement of the problem

Objective function

$$\text{Minimize } C_d\left[\left(\sum_{i\in City}\sum_{j\in Incinerator} D^x_{ij}X_{ij}\right)+\left(\sum_{j\in Incinerator}\sum_{k\in Landfill} D^y_{jk}Y_{kj}\right)\right] + \sum_{j\in Incinerator} C^b_j Z_j$$

i.e.,

$$\text{MIN } 3[(30X_{11}+5X_{12}+36X_{21}+42X_{22})+(5Y_{11}+8Y_{12}+9Y_{21}+6Y_{22})] + 40Z_1 + 30Z_2$$

i.e.,

```
MIN = TRANSPORTCOST + BURNCOST;
   TRANSPORTCOST = COSTPERMI*(@SUM(PICKUP: DX*X)
      + @SUM(DISPOSE: DY*Y) );
   BURNCOST = @SUM(INCINERATORS:ICOST*Z);
```


Computing travel distances

$$\sum_{i\in City}\sum_{j\in Incinerator} D^x_{ij}X_{ij}$$

Use @SUM to loop through the deliveries of trash to incinerators:

with or without indices:

```
@SUM(PICKUP(I,J):   DX(I,J)*X(I,J) )
```

or simply

```
@SUM(PICKUP:   DX*X )
```

so

```
MIN = TRANSPORTCOST + BURNCOST;
   TRANSPORTCOST = COSTPERMI*(@SUM(PICKUP: DX*X)
      + @SUM(DISPOSE: DY*Y) );
   BURNCOST = @SUM(INCINERATORS:ICOST*Z);
```

constraints

subject to

$$\sum_{j \in Incinerator} X_{ij} = OUTPUT_i \qquad \text{for all } i \in City$$

$$\left. \begin{aligned} Z_j &= \sum_{i \in City} X_{ij} \\ Z_j &\le I_j^{cap} \end{aligned} \right\} \quad \text{for all } j \in Incinerator$$

$$0.2 Z_j = \sum_{k \in Landfill} Y_{jk} \qquad \text{for all } j \in Incinerator$$

$$\sum_{j \in Incinerator} Y_{jk} \le L_k^{cap} \qquad \text{for all } k \in Landfill$$

$$X_{ij} \ge 0,\ \ Y_{jk} \ge 0,\ \ \&\ Z_j \ge 0 \qquad \text{for all } \ i, j, \& k$$

constraints

$$\sum_{j \in Incinerator} X_{ij} = OUTPUT_i \qquad \text{for all } i \in City$$

i.e.,

$$X_{11} + X_{12} = 500 \qquad \text{for } City\ 1$$
$$X_{21} + X_{22} = 400 \qquad \text{for } City\ 2$$

Use @FOR to generate a set of constraints for each city:

```
@FOR(CITIES(I):
   @SUM(INCINERATORS(J): X(I,J)) = OUTPUT(I)
);   ! be sure to close the parenthesis ;
```


$$\left. \begin{aligned} Z_j &= \sum_{i \in City} X_{ij} \\ Z_j &\le I_j^{cap} \end{aligned} \right\} \quad \text{for all } j \in Incinerator$$

i.e.,

$$\left. \begin{aligned} Z_1 &= X_{11} + X_{21} \\ Z_1 &<= 500 \end{aligned} \right\} \quad \text{for } Incinerator\ 1$$

$$\left. \begin{aligned} Z_2 &= X_{12} + X_{22} \\ Z_2 &<= 500 \end{aligned} \right\} \quad \text{for } Incinerator\ 2$$

i.e.,

```
@FOR(INCINERATORS(J):
   Z(J)=@SUM(CITIES(I): X(I,J) );
   Z(J) <= ICAP(J);
);
```


$$0.2 Z_j = \sum_{k \in Landfill} Y_{jk} \qquad \text{for all } j \in Incinerator$$

i.e.,

$$0.2\ Z_1 = Y_{11} + Y_{12} \qquad \text{for } Incinerator\ 1$$
$$0.2\ Z_2 = Y_{21} + Y_{22} \qquad \text{for } Incinerator\ 2$$

i.e.,

```
@FOR(INCINERATORS(J):
   REDUCTION* Z(J) = @SUM(LANDFILLS(K):  Y(J,K) );
);
```

$$\sum_{j \in Incinerator} Y_{jk} \le L_k^{cap} \qquad \text{for all } k \in \mathit{Landfill}$$

i.e.,

$Y_{11} + Y_{21} <= 200$ for *Landfill 1*

$Y_{12} + Y_{22} <= 200$ for *Landfill 2*

i.e.,

```
@FOR(LANDFILLS(K):
   @SUM(INCINERATORS(J): Y(J,K) ) <= LCAP(K)
);
```



```
   MIN = TRANSPORTCOST + BURNCOST;
   BURNCOST = @SUM(INCINERATORS:ICOST*Z);
   TRANSPORTCOST = COSTPERMI*(@SUM(PICKUP: DX*X)
      + @SUM(DISPOSE: DY*Y) );
   @FOR(CITIES(I):
      @SUM(INCINERATORS(J): X(I,J)) = OUTPUT(I)
   );
   @FOR(INCINERATORS(J):
      Z(J)=@SUM(CITIES(I): X(I,J) );
      Z(J) <= ICAP(J);
   );
   @FOR(INCINERATORS(J):
      REDUCTION* Z(J) = @SUM(LANDFILLS(K):  Y(J,K) );
   );
   @FOR(LANDFILLS(K):
      @SUM(INCINERATORS(J): Y(J,K) ) <= LCAP(K)
   );
END
```



```
Global optimal solution found at step:            4
 Objective value:                          84700.00

       Variable           Value        Reduced Cost
  TRANSPORTCOST        53700.00            0.00
       BURNCOST        31000.00            0.00
         Z( 1)           400.00            0.00
         Z( 2)           500.00            0.00
      X( 1, 1)             0.00           84.40
      X( 1, 2)           500.00            0.00
      X( 2, 1)           400.00            0.00
      X( 2, 2)             0.00            8.60
      Y( 1, 1)            80.00            0.00
      Y( 1, 2)             0.00            9.00
      Y( 2, 1)             0.00            9.00
      Y( 2, 2)           100.00            0.00
```



```
     Row    Slack or Surplus      Dual Price
      1         84700.00           1.000000
      2        0.0000000          -1.000000
      3        0.0000000          -1.000000
      4        0.0000000          -43.00000
      5         100.0000          0.0000000
      6        0.0000000           15.00000
      7        0.0000000          -33.60000
      8        0.0000000          0.0000000
      9        0.0000000           18.00000
     10        0.0000000          -48.60000
     11        0.0000000          -151.0000
     12         120.0000          0.0000000
     13         100.0000          0.0000000
```